
Ajuste o Medidor SEL-734 para Proteção e 
Controle de Bancos de Capacitores da Distribuição 

 

Jeff Anderson e Michael Thompson 

INTRODUÇÃO 

O Sistema de Medição de Faturamento do SEL-734 propicia medição de faturamento com alta 

precisão e medição da qualidade de energia para aplicações industriais e de empresas 

concessionárias de energia elétrica. O SEL-734 possui as equações de controle SELOGIC , 

flexíveis e programáveis pelo usuário, que incluem funções matemáticas. As funções de controle 

e medição são ideais para proteção e controle de bancos de capacitores.  

O SEL-734 pode também executar o controle automático de bancos de capacitores. Este guia de 

aplicação explica como configurar o SEL-734 para o controle automático baseado na tensão do 

sistema e na hora do dia. O guia aborda também a programação do SEL-734 para gerar alarme ou 

trip na presença de componentes harmônicos prejudiciais, usando as funções matemáticas 

incorporadas nas equações de controle SELOGIC. 

Um transformador de corrente no neutro de um banco de capacitores com conexão estrela 

aterrada pode ser ligado na entrada de corrente IN  do SEL-734. Usando as equações matemáticas 

SELOGIC, essa grandeza analógica de corrente pode ser usada para detectar queima de fusíveis no 

capacitor. Uma aplicação similar pode ser usada para um banco de capacitores com conexão 

estrela dual não aterrada.  

 

Figura 1: Diagrama Unifilar da Proteção e Controle de Banco de Capacitores SEL-734 

CONFIGURAR O SEL-734 PARA CONTROLE AUTOMÁTICO 

Controle Baseado na Tensão do Sistema 

A lógica usada para o controle automático de bancos de capacitores consiste na simples 

comparação de uma grandeza do sistema com dois valores limites ajustados. O exemplo usado 

neste guia é baseado no controle de tensão. Entretanto, somente é necessário um mínimo de 

esforço para substituir a tensão por outras grandezas do sistema, tais como VARs ou corrente, na 

mesma lógica.  

Para controlar o chaveamento de um banco de capacitores da distribuição baseando-se na tensão, 

a tensão medida é comparada com uma faixa de tensão desejada definida por dois valores limites. 

Quando a tensão cair abaixo do limite menor, o banco é chaveado, sendo ligado. Quando a tensão 

subir a um valor superior ao do limite maior, o banco é chaveado, sendo desligado. Para este 

exemplo, o limite da tensão baixa é ajustado em 90% da nominal e o limite da tensão alta é 

ajustado em 110% da nominal. Os valores limites reais dos ajustes da tensão devem ser baseados 

em um estudo do sistema para calcular a mudança esperada na tensão quando do chaveamento do 

banco de capacitores. A tensão secundária nominal é estimada em 120 V fase-neutro. O número 

de operações por dia é limitado em quatro, visando evitar que a lógica efetue o chaveamento 

continuamente, ligando e desligando o banco se a tensão não puder ser estabilizada entre os 

valores limites. 

Nota: Ao longo deste guia de aplicação, vamos mostrar o prompt dos ajustes seguido de 

informações que você pode inserir. Essas informações incluem o ajuste, seguido do 

significado do ajuste. Os dados que você inserir no medidor a partir de seu computador 

são apresentados em negrito. 

Primeiro, ajuste os valores limites e defina as variáveis de tensão: 

EMV := 6 

MV01 := 108 # LIMITE DA TENSAO BAIXA EM VOLTS SECUNDARI OS 

MV02 := 132 # LIMITE DA TENSAO ALTA EM VOLTS SECUNDARIOS  

MV03 := VA_MAG # MAGNITUDE DA TENSAO DA FASE A  

MV04 := VB_MAG # MAGNITUDE DA TENSAO DA FASE B  

MV05 := VC_MAG # MAGNITUDE DA TENSAO DA FASE C  

MV06 := MINSM # MINUTOS APOS A MEIA -NOITE  

Em seguida, compare as magnitudes de tensão com os limites. Programe a lógica para abrir ou 

fechar automaticamente o banco se os limites forem ultrapassados por um valor de tempo 

ajustado. A lógica inclui um intertravamento para evitar o fechamento durante cinco minutos após 

uma operação de abertura. Isto é para permitir que a carga residual dos capacitores seja 

descarregada. O tempo real necessário para descarregar os capacitores pode variar dependendo 

dos parâmetros do sistema. Se todas as três tensões estiverem abaixo do valor limite de subtensão 

de fase incorporado (determinado pelo ajuste 27P1P), a lógica é desabilitada: 

ESV := 6 

ESVT := 5 

ESC := 1 

Ajuste a lógica de fechamento.Ver Figura 2. 

SV01 := NOT (27A OR 27B OR 27C) # VERIFICA A TENSAO BOA 

SV02 := (MV03 < MV01 OR MV04 < MV01 OR MV05 < MV01) AND NOT 

SV05T AND NOT SC01QU AND SV01 # COMPARA AS TENSOES 

DE FASE COM O LIMITE MENOR AO MESMO TEMPO EM QUE 

SUPERVISIONA A TENSAO E O CONTADOR. AJUSTE O TEMPO 

DE PICKUP EM 5 SEGUNDOS E O TEMPO DE ñDURACAO DO 

TRIPò EM 0,5 SEGUNDO 

SV02PU := 5.000 

SV02DO := 0.500 

OUT101 := SV02T # LIGA O BANCO DE CAPACITORES 

Aguarde cinco minutos após a abertura para que o capacitor descarregue: 

SV04 := R_TRIG SV03T # DETECCAO DE OPERACAO DE ABERTURA DA 

CHAVE  

SV04PU := 0.000 

SV04DO := 250.000 

SV05 := SV04T 

SV05PU := 0.000 

SV05DO := 50.000 

Figura 2: Lógica de Fechamento por Controle de Tensão 

Limite o número de operações de fechamento a quatro por dia. Ver Figura 3: 

SC01PV := 4 

SC01R := R_TRIG SV06 # RESETA NA ATIVACAO D E SV06 

SC01LD := 0 

SV01CU := R_TRIG SV02T # DISPARA O CONTADOR NA OPERACÃO DE 

FECHAMENTO DA CHAVE  

SC01CD := 0 

SV06 := MV06 = 1 # RESETA O CONTADOR TODA NOITE AS 00:01 H 

(12:01 AM) 

 

Figura 3: Lógica do Contador do Número Máximo de Operações Por Dia 

 

Ajuste a lógica de abertura. Ver Figura 4: 

SV03 := MV03 > MV02 OR MV04 > MV02 OR MV05 > MV02  # 

COMPARA AS TENSOES DE FASE COM O LIMITE MAIOR  

SV03PU := 5.000 

SV03DO := 0.500 

OUT102 := SV03T # DESLIGA O BANCO DE CAPACITORES 



 

          SCHWEITZER ENGINEERING LABORATORIES, COMERCIAL LTDA. 

 

 

Rod. Campinas/Mogi-Mirim (SP340), Km 118,5 ï Pólis de Tecnologia ï Prédio 11 ï 13086-902 ï Campinas/SP 
Tel.: (19) 3515.2010 / Fax. (19) 3515.2011 - www.selinc.com.br ï suporte@selinc.com                 Pág. - 6/18 

 

Figura 4: Lógica de Abertura por Controle de Tensão 

Quando estiver usando tensão para controlar um banco de capacitores, tome cuidado na escolha 

dos valores limites para ñligarò (ñturn onò) e ñdesligarò (ñturn offò). Ao ligar o banco de 

capacitores, ocorrerá a elevação da tensão. Portanto, você deve normalmente ajustar a faixa entre 

o limite menor para ñturn onò e o limite maior para ñturn offò em 1,5 vezes o aumento de tens«o 

estimado calculado. De outra forma, o controle chaveará continuamente o banco, ligando e 

desligando até que o contador atinja o n¼mero quatro. Voc° pode usar o ñMeter Word bitò de 

saída SC01QU para gerar um alarme para esta condição. 

Controle Baseado na Hora do Dia 

O chaveamento automático de um banco de capacitores baseado na hora do dia pode também ser 

executado através de uma lógica ligeiramente diferente. Usando a grandeza MINSM, ou Minutos 

Após a Meia-Noite (ñMinutes Past Midnightò), ajuste o SEL-734 para ligar ou desligar o banco 

em um determinado horário do dia. A lógica a seguir vai configurar o SEL-734 para ligar o banco 

de capacitores às 16:00h (4:00 PM) e desligar às 20:00h (8:00 PM) nos dias úteis: 

EMV := 4 

MV01 := 960 # 960 MINUTOS APOS A MEIA-NOITE, OU 16:00H (4:00 PM) 

MV02 := 1200 # 1200 MINUTOS APOS A MEIA-NOITE, OU 20:00H (8:00 PM) 

MV03 := MINSM # VARIAVEL MINUTOS APOS A MEIA -NOITE  

MV04 := DAYW # NUMERO DO DIA DA SEMANA USADO PARA FINS DE 

SEMANA 

Compare a hora atual do dia com os tempos de chaveamento pré-ajustados: 

ESV := 4 

ESVT := 4 

SV01 := MV03 = MV01 

SV01PU := 0.000 

SV01DO := 0.000 

Ligar o banco de capacitores se for um dia útil. Ver Figura 5: 

SV03 := R_TRIG SV01 AND MV04 > 1 AND MV04 < 7 

SV03PU := 0.000 

SV03DO := 0.500 

OUT101 := SV03T # LIGA O BANCO DE CAPACITORES 




